
· If your transaction executes database updates from within the dialog program, you must bundle all of your database updates into a single dialog step (usually the last).  This is the only way to ensure that your database changes are processed using the "all or nothing" principle.

· If you have updates executed from the dialog program, you must save the data you want to change in the global program data until the database changes are made. This data is written to the database with the status it had for the last dialog step.

· If you update the database directly from within the program, your program must set and release SAP locks itself. Use the specified sequence of steps (see the unit SAP Locking Concept).

1. Lock data
2. Read data (+ display for change)
3. Change data in database
4. Unlock data

· Remember that your program must delete the lock entries.  For this purpose you can execute either the unlock module   DEQUEUE__<lock object>, which belongs to the lock object, or the general unlock function module DEQUEUE_ALL.

· Database updates from dialog mode can be executed in bundled form by using the special subroutine technique PERFORM <subroutine> ON COMMIT.

· The statement PERFORM <subroutine> ON COMMIT registers the subroutine that has been specified.  This will not be executed until the system reaches the next COMMIT WORK statement. 

· If the database updates are encapsulated in the subroutines, they can be separated from the program logic and relocated to the end of the LUW processing.

· Each subroutine registered with PERFORM ON COMMIT is executed only once per LUW.  Calls can be made more than once (no errors); the subroutine, however, is executed only once.

· Nested PERFORM ON COMMIT calls after release 4.6 trigger a runtime error.

· The COMMIT WORK statement carries out all subroutines registered  to be executed, one after the other, and triggers a database commit.
· If there is an error, you can terminate processing from the respective subroutine with a dialog message A (MESSAGE A); the previous consistent database status can be restored with a type A dialog message and the previous consistent database status can be restored.  

· Subroutines called using PERFORM ON COMMIT must have no interface.  They work instead with global data, that is, with the values the data objects contain at the point when the subroutine is actually run.

· The PERFORM ON COMMIT technique can also be used in the update. This will be discussed later.

· For further information, see the ABAP Editor keyword documentation for PERFORM.

· Update techniques allow you to separate user dialogs from the database changes. Both are executed by different programs, which generally run in different work processes.

· The user uses a dialog program to execute dialogs. A dialog program accepts user entries and, at the end, when it passes on the changed data, triggers an update program that updates the corresponding data in the database.  

· No dialogs run in the update programs.

· The following slides describe the various steps in a program that uses an update technique.

· Step 1: The dialog program receives the data updated by the user and writes it to a special log table. The entries in this table are of a request type. The data contained in them will be written to the database later by program.

· A dialog program can write several entries to the log table.

· The entries in the log table represent an LUW, in other words, they will either be executed in the database together or not at all ("all or nothing" principle).

· Step 2: The dialog program closes the logical data packet (LUW) that was written to the log table. It also informs the Basis system that there is a packet to update.

· Step 3: A Basis program reads the data associated with the LUW from the log table and supplies it to the update program.

· Step 4: The update program accepts the data transferred to it and updates the database entries.

· Step 5: If the update program runs successfully, a Basis program deletes all entries for the LUW from the log table.

· In the event of an error, the entries remain in the log table. They are flagged as incorrect.  The user who has triggered the update is generally informed by express e-mail about the error.

· You can set the option of informing users by e-mail that an update action has failed using the profile parameters rdisp/vbmail and rdisp/vb_mail_user_list. . 

· The parameter rdisp/vbmail can be given the value 1 or 0 (mail dispatch if error or not).

· The value given to rdisp/vb_mail_user_list defines who is to be contacted if there is an error.   The value $ACTUSER means that the user who has created the data record to be updated will be informed. 

· Transaction SM13 is used to monitor update requests.

· Technical implementation of the update concept requires, in addition to the program that monitors the user dialog, an update program that must be implemented as a special function module (update module). 

· Update modules, like other function modules, have an interface for transferring data. The interface for update function modules only includes IMPORTING and TABLES parameters. These must be typed using reference fields or structures. 

· Export parameters and exceptions are ignored in update modules.

· The function module contains the actual database update statements. 

· You create update requests from the dialog program by calling the respective update function module.  Use the IN UPDATE TASK addition.  This means that the function module is not executed immediately, but is written to the log table, together with the input data, as an execution request.

· All of the update requests in an SAP LUW are stored under the same update key.  The update key is a unique worldwide identification code for an SAP LUW.

· Only when the system finds a COMMIT WORK statement will it create a header entry for the requests that belong together (log header). Then the system closes the unit. The log header contains information on the dialog program that wrote the log entries and information on the update modules to be executed.

· After the log header is created, the system informs the dispatcher that an update package is ready for processing.

· Sometimes you may need to discard all change requests that were written up to now for the current SAP LUW.   This is the case, for example, when you terminate the transaction.

· To discard the current SAP LUW during the generation phase, use the ABAP statement ROLLBACK WORK or send a type A dialog message.  Both procedures
 - Delete all change requests written up to that point
 - Delete all the locks set up to that point
 - Discard all the changes executed in the current DB LUW
 - Discard all the form routines registered using PERFORM ON COMMIT.

· The ROLLBACK WORK statement does not affect the program context. In other words, all data objects (program-specific objects and objects from function groups that may be used) remain unchanged, and they are not reset.

· The task of an update module is to pass the requests for database updates to the database and to evaluate their return codes.

· If you want to trigger a database rollback in the update module, issue a type A dialog message. This also terminates the processing of the current SAP LUW. The log entry belonging to the SAP LUW is flagged as containing an error. The termination message is also entered in the log.

· You can examine the log entry using transaction SM13. 

· The system sends an express mail to the relevant user, telling him or her that the LUW update was terminated.  For this purpose, you must set the profile parameters rdisp/vbmail and rdisp/vb_mail_user_list accordingly:

· The parameter rdisp/vbmail can be given the value 1 or 0 (mail dispatch if error or not).

· The value given to rdisp/vb_mail_user_list defines who is to be contacted if there is an error.   The value $ACTUSER means that the user who has created the data record to be updated will be informed. 

· You may not use the explicit ABAP statements COMMIT WORK or ROLLBACK WORK in an update module.

· If you have locks set in a dialog program that works with the update technique and these locks have been set using _SCOPE = 2, you can pass these on to the update task at COMMIT WORK. After this, they are no longer accessible by the dialog program.

· You do not need to release the locks explicitly in the update modules because a Basis program at the end of the update process automatically releases the locks.

· This automatic release of locks by the update task will also take place if there is an error, that is, if one of the update modules terminates and rolls back LUW processing through the output of a termination dialog message.

· If the update modules allow failed update requests to be reprocessed (see V1 update), the database in the database tables at the point of reprocessing may be different from that at the point of the failed update attempt. Also, reprocessing generally takes place without locks because locks are automatically deleted by the update task if there is an error.

· For example, reprocessing is appropriate if a document was not written successfully to the database because of an overflow of table spaces. 

· In asynchronous updates, the dialog program and update program run separately.

· The dialog program writes the change requests to the log table and closes the LUW with a COMMIT WORK.

· The update initiated by the COMMIT WORK now processes the change requests.  The dialog program is continued; the system does not wait for the update to end.

· The update program runs in a special update work process. This can be on an application server other than the one used for the SAP R/3 System.

· Asynchronous updates are useful in transactions where the database updates take a long time and the perceived performance by the shorter user dialog response time is important.

· Asynchronous updating is the standard technique used in dialog processing.

· You can implement the log table VBLOG as a cluster file in your system, or replace it with the transparent tables VBHDR, VBMOD, VBDATA, and VBERROR.

· If you have a synchronous update that is triggered by COMMIT WORK AND WAIT, the dialog program waits for the update to end before program processing is continued. 
· You use this update mode if further processing or dialog program terminationdepends on the update result. 

· You can query the processing success of the synchronous update using the system field sy-subrc.

· During the entire waiting phase, the dialog program is in a "rolled out" state. This means that the respective dialog work process is released for further use.  When the update is complete, the system again assigns a free dialog work process for further processing. 

· In local updates, the update functions are run in the same dialog process used by the dialog program containing the dialog program (that is, locally). Processing of the dialog program is continued when the update is completed (synchronous).

· To have update modules executed locally, you must use the SET UPDATE TASK LOCAL statement before you write the respective requests. Close the written requests with the COMMIT WORK statement, and these will be processed in the same dialog work process.

· After the local update has been processed successfully, a DB commit is initiated explicitly and the dialog program is continued.

· If there is an error and a termination message is dispatched by one of the update modules, the system executes an automatic DB rollback to discard the changes in the current LUW and the dialog program is terminated by the display of a termination message. 

· In the local update mode, change requests are not written to the database table VBLOG, but kept in main memory. Due to the missing IO accesses, this is quicker than for synchronous or asynchronous updates.   The disadvantage, however, lies in the exclusive use of a work process.   Therefore, local updates are appropriate in batch mode only.

· SET UPDATE TASK LOCAL is possible only if no requests were created for the current LUW. This statement works only until the next COMMIT WORK.

· There are two types of update modules: V1 and V2. The type of update module determines its processing mode. To begin with, all V1 requests in the dialog program are executed as independent DB LUW. Only if the V! requests are executed successfully are V2 requests processed, also as independent LUWs (thus V1 then V2 update phase).

· V2 update modules are used for database changes that are linked to the V1 changes (main changes) but do not necessarily have to be executed in the same DB LUW (for example, updating of statistics).

· V1 modules can be restartable or non-restartable. If an update error occurs, you can manually restart requests that were created by restartable update modules using transaction SM13. You do this after you have cleaned up the application error in question. V2 update modules can always be restarted for processing if there was an error.

· The classification collective run is used internally by SAP for V2 modules.  It is a special type of V2 update.   Corresponding requests are not executed directly after the V1 update, but only after the collector program RSM13005 (generally planned ahead) has been called.

· The flow diagrams displayed in the previous sub-units always show V1 requests that were created by V1 update modules.

· V1 update modules create requests in the VBLOG table for synchronous and asynchronous updates.  For local updates, the V1 requests are retained in the main memory.

· V2 requests are always stored in the VBLOG table.

· V1 requests are processed in a V1 update work process as an independent DB LUW.  If the V1 update is successful, the system deletes the V1 requests and all the locks passed on to the update task, sets a DB Commit, and triggers the V2 update.  

· V2 requests are executed in a V2 update work process. These, too, form an independent DB LUW. If no V2 update work process is set up in SAP R/3, V2 requests are executed in a V1 update work process. If all the V2 requests were processed successfully, they are removed from the VBLOG table and the system also sets a DB commit. V2 requests generally run without locks because they were already deleted after completion of the V1 update. 

· If the V1 update module terminates the V1 update with a termination message, all the locks passed on to the update task are deleted, a DB rollback takes place, an e-mail is sent to the user who created the LUW, and the V1 requests are flagged as incorrect in the VBLOG with the corresponding termination message. The V2 update is not triggered.

· However, if the V2 update is terminated by a termination message of a V2 update module, the system triggers a database rollback.  All of the V2 changes in the SAP LUW are undone and the V2 requests in VBLOG are flagged as incorrect with the corresponding termination message.

· The locks created from within the dialog program using _SCOPE = 2 (default) are transferred to the V1 update task at COMMIT WORK (AND WAIT). At the end of the V1 update, they are automatically deleted, regardless of whether the V1 update was successful or whether the update was terminated by an error with issue of a termination message. 

· Therefore, lock entries must not be explicitly removed either in the dialog program (too early) or in the update module (unnecessary).

· The V2 update always runs without SAP locks.

· Each time a change is updated to the database, the record to be changed is physically locked by the database up to the end of the current database LUW (DB commit or DB rollback).  The same applies if you are reading with SELECT ... FOR UPDATE.

· However, it is a good idea, for performance reasons, to keep such automatically set database locks as short as possible. This is because read accesses to the respective record are not allowed for the duration of the physical lock (they must wait until the record is released by the database) and many programs execute read access without locks.. 

· For this you should adhere to the following rules when programming inline changes and update modules:

· Create new table entries first.  Their database locks pose the least "interference" for other users.

· Perform table updates that are not critical to performance. As a rule, these tables are accessed simultaneously by relatively few users.

· Always change tables that represent central resources in the system (tables accessed by several users simultaneously) late within an LUW, if possible, so that the respective database locks hinder others only for a short time.

· During the update, the goal is to execute changes to central tables (performance-critical, often accessed simultaneously by several users) as late as possible in the LUW. To achieve this, you can use the PERFORM ON COMMIT technique in the update.

· Encapsulate the changes to central tables as form routines within the appropriate function group that belongs to the corresponding update module. Then call the routines from within the update module using PERFORM ON COMMIT. The form routines are then not executed until the last update module has been processed.

· Remember that these form routines must work with the global data of the respective function group.

	Unit: Organizing Database Updates

	At the conclusion of these exercises, you will be able to:

· Perform database updates using the asynchronous update technique

	The program SAPMZ##_BOOKINGS2 from the previous unit is to be changed or enhanced so that database updates can be performed using the asynchronous update technique. 

Canceling bookings:
To implement the asynchronous update technique, you must adjust the existing source code.

Creating a new booking:
The database dialog part is to be implemented here. The data for a new booking is entered on screen 300. Clicking the Save icon (function code SAVE) on screen 300 will insert the new bookings in the SBOOK table and modify the flight in question in the SFLIGHT table. The updates are to be performed within a DB LUW and using the asynchronous update technique.

	Program:

SAPMZ##_BOOKINGS3
Transaction code: 
Z##_BOOKINGS3
Template:
SAPBC414T_BOOKINGS_03
Model solution:
SAPBC414S_BOOKINGS_03


1-1
Copy your solution SAPMZ##_BOOKINGS2 or the program template SAPBC414T_BOOKINGS_03 with all subobjects to SAPMZ##_BOOKINGS3 (## is your group number).
Assign transaction code Z##_BOOKINGS3 to the program.

1-2
Canceling existing bookings: 

1-2-1
Function modules UPDATE_SFLIGHT and UPDATE_SBOOK are used to update the table entries in the DB tables SFLIGHT and SBOOK.  Can these function modules also be used to perform the updates using the update technique?

1-2-2
Modify your program so that the updates to the DB tables SFLIGHT and SBOOK  are performed using the update technique:

· Call the corresponding function modules capable of performing updates in the SAVE_MODIFIED_BOOKING subroutine.

· Insert the COMMIT WORK statement in the PAI module USER_COMMAND_0200.
· Also, be aware of the fact that the locks set using "_SCOPE = 2" are passed to the update program and must not, therefore, be explicitly released in the dialog program.

1-3
Generating a new booking:
To generate a new entry in the DB table SBOOK, use the function module  INSERT_SBOOK, which is capable of performing updates.  This function module is to be called up in the subroutine SAVE_NEW_BOOKING.  The subroutine is called up from the PAI module USER_COMMAND_0300 (screen 300) and is already created (blank).

1-3-1
If you have copied your previous solution program SAPMZ##_BOOKINGS2 as a template for this exercise, you should – when the subroutine SAVE_NEW_BOOKING starts – copy the subroutine CONVERT_TO_LOC_CURRENCY, and pass on  WA_SBOOK. This is necessary so that you can fill the corresponding currency fields in WA_SBOOK:
PERFORM convert_to_loc_currency USING wa_sbook.
1-3-2
Call up the function modules INSERT_SBOOK and UPDATE_SFLIGHT, which are capable of performing updates, to update the DB tables SBOOK and SFLIGHT using the update technique.

1-3-3
Insert the COMMIT WORK statement in the PAI module USER_COMMAND_0200.
1-3-4
Lock the flight and the booking by calling up the corresponding lock modules.  Call up subroutine ENQ_SFLIGHT and ENQ_SBOOK in the appropriate places. 
If a user action calls up screen 100, release the locks.

	The booking data is held in structure WA_SBOOK.

	Unit: Organizing Database Updates


Model Solution SAPBC414S_BOOKINGS_03
1 PAI Modules
*------------------------------------------------------------------*

***INCLUDE BC414S_BOOKINGS_03I01 .
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Module  EXIT  INPUT
*&-----------------------------------------------------------------*

MODULE exit INPUT.
  CASE ok_code.
    WHEN 'CANCEL'.
      CASE sy-dynnr.
        WHEN '0100'.
          LEAVE PROGRAM.
        WHEN '0200'.
          PERFORM deq_all.
          LEAVE TO SCREEN '0100'.
        WHEN '0300'.
* remove all database locks
          PERFORM deq_all.
          LEAVE TO SCREEN '0100'.
        WHEN OTHERS.
      ENDCASE.
    WHEN 'EXIT'.
      LEAVE PROGRAM.
    WHEN OTHERS.
  ENDCASE.
ENDMODULE.                             " EXIT  INPUT
*&-----------------------------------------------------------------*

*&      Module  USER_COMMAND_0100  INPUT
*&-----------------------------------------------------------------*

MODULE user_command_0100 INPUT.
  CASE save_ok.
****************************CANCEL BOOKING**************************
    WHEN 'BOOKC'.
      PERFORM enq_sflight_sbook.
      PERFORM read_sflight USING wa_sflight sysubrc.
      PERFORM process_sysubrc_bookc.
      PERFORM read_spfli USING wa_spfli.
      PERFORM read_sbook USING itab_book itab_cd.
      REFRESH CONTROL 'TC_SBOOK' FROM SCREEN '0200'.
****************************CREATE BOOKING**************************
    WHEN 'BOOKN'.
* lock flight in Table SFLIGHT, which will be modified when new 
* booking is saved
      PERFORM enq_sflight.
      PERFORM read_sflight USING wa_sflight sysubrc.
      PERFORM process_sysubrc_bookn.
      PERFORM read_spfli USING wa_spfli.
      PERFORM initialize_sbook USING wa_sbook.
    WHEN 'BACK'.
      SET SCREEN 0.
    WHEN OTHERS.
      SET SCREEN '0100'.
  ENDCASE.
ENDMODULE.                             " USER_COMMAND_0100  INPUT
*&-----------------------------------------------------------------*

*&      Module  USER_COMMAND_0200  INPUT
*&-----------------------------------------------------------------*

MODULE user_command_0200 INPUT.
  CASE save_ok.
    WHEN 'SAVE'.
      PERFORM collect_modified_data USING itab_sbook_modify.
      PERFORM save_modified_booking.
* start asynchronous update and new SAP-LUW
      COMMIT WORK.
* database locks should not be removed here since they will be
* removed by the update program later.
      SET SCREEN '0100'.
    WHEN 'BACK'.
      PERFORM deq_all.
      SET SCREEN '0100'.
    WHEN OTHERS.
      SET SCREEN '0200'.
  ENDCASE.
ENDMODULE.                             " USER_COMMAND_0200  INPUT
*&-----------------------------------------------------------------*

*&      Module  USER_COMMAND_0300  INPUT
*&-----------------------------------------------------------------*

MODULE user_command_0300 INPUT.
  PERFORM tabstrip_set.
  CASE save_ok.
    WHEN 'NEW_CUSTOM'.
      PERFORM create_new_customer.
      SET SCREEN '0300'.
    WHEN 'SAVE'.
      PERFORM save_new_booking.
* start asynchronous update and new SAP-LUW
      COMMIT WORK.
* database locks will be removed by update program
      SET SCREEN '0100'.
    WHEN 'BACK'.
* remove all database locks
      PERFORM deq_all.
      SET SCREEN '0100'.
    WHEN OTHERS.
      SET SCREEN '0300'.
  ENDCASE.
ENDMODULE.                             " USER_COMMAND_0300  INPUT
2 FORM Routines
3 F01
*------------------------------------------------------------------*

***INCLUDE BC414S_BOOKINGS_03F01 .
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Form  PROCESS_SYSUBRC_BOOKN
*&-----------------------------------------------------------------*

FORM process_sysubrc_bookn.
  CASE sysubrc.
    WHEN 0.
      SET SCREEN '0300'.
    WHEN OTHERS.
* remove all database locks
      PERFORM deq_all.
      MESSAGE e023 WITH sdyn_conn-carrid sdyn_conn-connid
                        sdyn_conn-fldate.
  ENDCASE.
ENDFORM.                               " PROCESS_SYSUBRC_BOOKN
4 F04
*------------------------------------------------------------------*

*   INCLUDE BC414S_BOOKINGS_03F04
*------------------------------------------------------------------*

*&-----------------------------------------------------------------*

*&      Form  SAVE_MODIFIED_BOOKING
*&-----------------------------------------------------------------*

FORM save_modified_booking.
  CALL FUNCTION 'UPDATE_SBOOK' IN UPDATE TASK
       EXPORTING
            itab_sbook = itab_sbook_modify.
* no exception handling when using asynchronous update technique
  PERFORM update_sflight.
ENDFORM.                               " SAVE_MODIFIED_BOOKING
*&-----------------------------------------------------------------*

*&      Form  UPDATE_SFLIGHT
*&-----------------------------------------------------------------*

FORM update_sflight.
  CALL FUNCTION 'UPDATE_SFLIGHT' IN UPDATE TASK
       EXPORTING
            carrier    = wa_sflight-carrid
            connection = wa_sflight-connid
            date       = wa_sflight-fldate.
* no exception handling when using asynchronous update technique
ENDFORM.                               " UPDATE_SFLIGHT
*&-----------------------------------------------------------------*

*&      Form  SAVE_NEW_BOOKING
*&-----------------------------------------------------------------*

FORM save_new_booking.
  PERFORM convert_to_loc_currency USING wa_sbook.
* lock booking on DB table sbook to be created
  PERFORM enq_sbook.
  CALL FUNCTION 'INSERT_SBOOK' IN UPDATE TASK
       EXPORTING
            wa_sbook = wa_sbook.
* no exception handling when using asynchronous update technique
  PERFORM update_sflight.
ENDFORM.                               " SAVE_NEW_BOOKING

